Newsgroup: it.comp.appl.access
Titolo: selezionare record in maschera continua

Ho letto (ed ho provato ad eseguire) il suggerimento di 
Karl Donaubauer che è il seguente:
Dim ctl As Control
For Each ctl In Me.Controls
Select Case ctl.ControlType
Case acTextBox, acComboBox, acListBox
With ctl
.FormatConditions.Delete
.FormatConditions.Add acExpression, acEqual, "id=" & Me!id
.FormatConditions(0).BackColor = vbGreen
End With
End Select
Next ctl
sarà certamente qualcosa di molto semplice, ma ultimamente  i miei occhi sono foderati di prosciutto. Esistono nella form i campi idpaziente e id_esami. Ho provato a modificare il listato sostituendo ad id e me!id   i miei campi ma niente... potreste illuminarmi su dove non vedo la luce?
 
Grazie BC

Salve,
 
Cesare Briglia:

Come lo provi e cosa succede? Niente, qc che non vuoi o quale errore...?
 
Potrebb'essere circa:
 
..."idpaziente=" & Me!idpaziente & " And id_esami=" & Me!id_esami
 
-- 
Ciao
Karl
*********
http://AccessDevCon.com
 
Access FAQ: 
https://www.donkarl.com/it
 

ho così modificato il codice (e colore)
Dim ctl As Control
For Each ctl In Me.Controls
Select Case ctl.ControlType
Case acTextBox, acComboBox, acListBox
With ctl
.FormatConditions.Delete
.FormatConditions.Add acExpression, acEqual, "idpaziente=" & Me!IDPaziente
.FormatConditions(0).BackColor = vbYellow
End With
End Select
Next ctl
(il secondo id non era presente... mio errore)
adesso funziona, 
grazie ancora .
BC
 
ps prima di implementare questo codice  era presente un'altra FC che colorava le textbox a seconda del sesso dei pazienti (blu maschietti, rosa femminucce). ora è sparito tutto. Come posso fare?
 

 ps prima di implementare questo codice  era presente un'altra FC che colorava le textbox a seconda del sesso dei pazienti (blu maschietti, rosa femminucce). ora è sparito tutto. Come posso fare?
 
Se vuoi mantenere l'altra formattazione già esistente, potresti provare 
di fare tutto nel codice. In principio circa:
 
With ctl
 
     .FormatConditions.Delete
 
     .FormatConditions.Add _
     	acExpression, acEqual, "idpaziente=" & Me!idpaziente & _
			" And id_esami=" & Me!id_esami
     ctl.FormatConditions(0).BackColor = vbYellow
 
     ctl.FormatConditions.Add _
     	acExpression, acEqual, "maschietti = True"
     ctl.FormatConditions(1).BackColor = vbBlue
 
     ctl.FormatConditions.Add _
     	acExpression, acEqual, "femminucce = True"
     ctl.FormatConditions(2).BackColor = TuoValorePerRosa
 
Next ctl


Scusa il ritardo nel risponderti e continuo a ringraziarti per l'attenzione (lavoro in ospedale e attualmente non stiamo passando un periodo facile...)
questo il codice modificato come mi hai suggerito
 
Dim ctl As Control
For Each ctl In Me.Controls
Select Case ctl.ControlType
Case acTextBox
With ctl
    .FormatConditions.Delete
    .FormatConditions.Add acExpression, acEqual, "idpaziente=" & Me!IDPaziente 
    ctl.FormatConditions(0).BackColor = vbYellow
 
    ctl.FormatConditions.Add acExpression, acEqual, Me.Sesso = "M"
    ctl.FormatConditions(1).BackColor = vbBlue
 
    ctl.FormatConditions.Add acExpression, acEqual, Me.Sesso = "F"
    ctl.FormatConditions(2).BackColor = vbMagenta
End With
End Select
Next ctl
 
però ora il record selezionato è giallo e TUTTI gli altri sono blu o magenta a seconda del sesso del paziente  selezionato...
Grazie ancora 
BC

 però ora il record selezionato è giallo e TUTTI gli altri sono blu o magenta a seconda del sesso del paziente  selezionato...
 
With ctl
     .FormatConditions.Delete
     .FormatConditions.Add acExpression, acEqual, "idpaziente=" & 
Me!IDPaziente
     .FormatConditions(0).BackColor = vbYellow
 
     .FormatConditions.Add acExpression, acEqual, "Sesso = 'M'"
     .FormatConditions(1).BackColor = vbBlue
 
     .FormatConditions.Add acExpression, acEqual, "Sesso = 'F'"
     .FormatConditions(2).BackColor = vbMagenta
End With
 
Meglio così?


 Access FAQ: 
https://www.donkarl.com/it
 
Perfetto!
Se mai dovessero fare una statua che rappresenti la gentilezza e la disponibilità, questa dovrebbe avere le fattezze di Karl Donaubauer.
Grazie ancora
 
BC